
2 Correction to UTRAN MDT testcase 8.6.2.1
2.1 f_TC_8_6_2_1_UTRAN() 
	Object name 
	f_TC_8_6_2_1_UTRAN()

	Reason for change
	In the current TTCN implementation, the function f_TC_8_6_2_1_UTRAN-> f_UTRAN34_PreambleGoToState6_10_MO-> f_UTRAN34_UE_PsRB_Est is called. Within this function, following is called :

v_NoOfPDPContexts := f_UTRAN_Global_IP_NoOfPDPContexts_Increment();
f_UTRAN_Global_IP_PDPInfo_Set((v_NoOfPDPContexts-1), true, v_PDN);

This causes a runtime error, because the above mentioned functions f_UTRAN_Global_IP_NoOfPDPContexts_Increment and f_UTRAN_Global_IP_PDPInfo_Set try to update the variables vc_UTRAN_Global_IP.NoOfPDPContexts and vc_UTRAN_Global_IP.PDPInfo  These are is not yet initialised.

	Summary of change
	Assigned  vc_UTRAN_Global_IP.NoOfPDPContexts := 0;  and vc_UTRAN_Global_IP.PDPInfo := omit

	Module
	UTRAN34_Logged_MDT.ttcn

	Label
	WA#


Before:
	function f_TC_8_6_2_1_UTRAN() runs on UTRAN_IP_PTC
  {
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    // Assign LAC and RAC for Cell2 according to Table 8.6.2.1-1
    f_UTRAN_CellInfo_SetLAC(utran34_Cell2, tsc_LAC_2);   // @sic R5-131756 sic@
    // Cell2 started switched off
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell2);
    f_UTRAN_SetCellPower(utran34_Cell2, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_Cell2);
    //Preamble to set UE to state 6-10 ( PS-DCCH+DTCH_DCH )
    f_UTRAN34_PreambleGoToState6_10_MO(utran34_Cell1);
    // Switch on cells
    f_UTRAN_SetCellPower(utran34_Cell2, -69, -69);
    f_UTRAN_TestBody_Set(true);
    fl_TC_8_6_2_1_Body();
    f_UTRAN_TestBody_Set(false);
    // Release RRC connection on cell 2 and release the cells
    f_UTRAN34_Postamble(utran34_Cell2, U2_CONNECTED);
    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));
  }


After:
	function f_TC_8_6_2_1_UTRAN() runs on UTRAN_IP_PTC
  {
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    //WA#8_6_2_1 : Changes needed to initialise the IP parameters need to be implemented.
    vc_UTRAN_Global_IP.NoOfPDPContexts := 0; // @sic R5-160730 sic@
    vc_UTRAN_Global_IP.PDPInfo := omit;
    //WA#8_6_2_1 : Change stop    
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    // Assign LAC and RAC for Cell2 according to Table 8.6.2.1-1
    f_UTRAN_CellInfo_SetLAC(utran34_Cell2, tsc_LAC_2);   // @sic R5-131756 sic@
    // Cell2 started switched off
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell2);
    f_UTRAN_SetCellPower(utran34_Cell2, tsc_AttenuationCellOFF, tsc_AttenuationCellOFF);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_Cell2);
    //Preamble to set UE to state 6-10 ( PS-DCCH+DTCH_DCH )
    f_UTRAN34_PreambleGoToState6_10_MO(utran34_Cell1);
    // Switch on cells
    f_UTRAN_SetCellPower(utran34_Cell2, -69, -69);
    f_UTRAN_TestBody_Set(true);
    fl_TC_8_6_2_1_Body();
    f_UTRAN_TestBody_Set(false);
    // Release RRC connection on cell 2 and release the cells
    f_UTRAN34_Postamble(utran34_Cell2, U2_CONNECTED);
    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));
  }


2.2 fl_TC_8_6_2_1_Body() 
	Object name 
	fl_TC_8_6_2_1_Body()

	Reason for change
	In the current TTCN implementation :
1) The function f_UTRAN_PrepareCellRRC_ConnEst is called with only utran34_Cell2 being passed as a parameter. This results in the cell state of the utran34_Cell1 being unchanged, which is incorrect. This function should be called with both the arguments. Please see screenshot.
2) The current TTCN implementation for capturing the UE Information Response is implemented such that it expects the UE Information Response message on SRB2. This is incorrect. The UE Information Response will always come on SRB4. This needs to be changed. Also, in the template cr_UEInformationResponse_LoggedMeasReportDef, * needs to be passed as the fourth argument for noncriticalextensions IE.

	Summary of change
	1) Instead of calling f_UTRAN_PrepareCellRRC_ConnEst(utran34_Cell2);, called f_UTRAN_PrepareCellRRC_ConnEst(utran34_Cell2, utran34_Cell1);

2) Changed the SRB from SRB2->SEB4 and included * as argument for noncriticalextensions in the UE Information response message. Please see screenshot. 

	Module
	UTRAN34_Logged_MDT.ttcn

	Label
	WA#


Before:
	function fl_TC_8_6_2_1_Body() runs on UTRAN_PTC
  {
    var float v_WaitToAllowToActivateLogging := 5.0;
    var B28_Type v_CellId;
    var B48_Type v_AbsoluteTimeInfo;
    var UTRAN_SecurityInfo_Type     v_SecurityInfo;
    var UTRAN_CellInfo_Type v_CellInfo;
    var PLMN_Identity v_PLMN_Identity;
    var UTRAN_FDD_TDD_Type v_UTRAN_FDD_TDD_Type := f_IRAT_GetUTRAN_FDD_TDD_Mode();
    var CellParametersID v_CellParametersID;
    var UTRAN_CellInfo_Type_FDD v_CellInfoFDD;
    var template (present) LoggedMeasInfoList v_LogMeasInfoList;
    v_CellId := f_UTRAN34_GetCellIndentity(utran34_Cell1);
    v_SecurityInfo := f_UTRAN_Security_Get();
    v_CellInfo := f_UTRAN_CellInfo_Get(utran34_Cell1);
    v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_Cell1);
    v_AbsoluteTimeInfo := f_MDT_GetAbsolute_TimeStamp();
    //@siclog "Step 1" siclog@
    //SS transmits a LoggedMeasurementConfiguration message to configure the UE to perform logging of measurement results while in Cell_PCH.
    U_AM.send(cas_RLC_Data_Req_NoCnf(utran_CellDedicated, tsc_RB2, cs_LoggingMeasurementConfigurationDef(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                                                                         v_PLMN_Identity,
                                                                                                         v_AbsoluteTimeInfo)));
    //@siclog "Step 2,3" siclog@
    //Procedure to release RRC Connection
    f_UTRAN34_RRC_ConnRel(utran34_Cell1, cell_Dch);
    //@siclog "Step 4" siclog@
    //Wait 5 seconds for UE logging interval timer to expire at least once
    f_Delay(v_WaitToAllowToActivateLogging);
    //@siclog "Step 5" siclog@
    //SS changes cell power levels accroding to T2 in the Table 8.6.2.1-1
    f_UTRAN_SetCellPower(utran34_Cell1, -69, -69);
    f_UTRAN_SetCellPower(utran34_Cell2, -60, -60);
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_Cell2); //@sic R5s160060 sic@
    //@siclog "Step 6" siclog@
    //The UE re-selects to Cell2 and performs Routing area update procedure. After completion of this procedure the UE is brought to idle mode
    f_UTRAN34_GMM_RAU(utran34_Cell2,?); // @sic R5s160060, R5-160925 sic@
    // Release RRC connection with cell2
    //f_UTRAN34_RRC_ConnRel(utran34_Cell2, cell_Dch); @sic R5s160060 sic@
    //@siclog "Step 7-10B" siclog@
    //Page UE to trigger RRC Connectin establishment
    //Complete RRC Connection setup and verify if IE "Logged Meas Available" is present
    f_UTRAN34_PS_MTCallEst(utran34_Cell2, true); //@sic R5s160060 sic@
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 10");    //@sic R5s160060 sic@
    //@siclog "Step 11" siclog@
    //Send UE INFORMATION REQUEST message
    v_SecurityInfo := f_UTRAN_Security_Get();
    U_AM.send(cas_RLC_Data_Req_NoCnf(utran_CellDedicated, tsc_RB2, cs_UEInformationRequest_LoggedMeasReportRequest(v_SecurityInfo.dl_IntegrityCheckInfo)));
    //@siclog "Step 12" siclog@
    // Receive UE INFORMATION RESPONSE message with Logged Meas Report
    select (v_UTRAN_FDD_TDD_Type) {
      case (UTRAN_FDD) {
        v_CellInfoFDD := f_UTRAN_CellInfo_GetFDDSpecific(utran34_Cell2);
        v_LogMeasInfoList := cr_LogMeasInfoList_FDD({permutation (cr_LogMeasInfo_FDD(complement(0),
                                                                                     cr_LoggedMeasServingCellMeas_FDD(v_CellId),
                                                                                     cr_LoggedMeasNeighbourInfoList_FDD(cr_PrimaryCPICH_Info(v_CellInfoFDD.priScrmCode))), *)});
      }
      case (UTRAN_TDD) {
        v_CellParametersID := f_UTRAN_CellInfo_GetCellParametersID_TDD(utran34_Cell2);
        v_LogMeasInfoList := cr_LogMeasInfoList_TDD128({permutation (cr_LogMeasInfo_TDD128(complement(0),
                                                                                           cr_LoggedMeasServingCellMeas_TDD128(v_CellId),
                                                                                           cr_LoggedMeasNeighbourMeasList_TDD128(v_CellParametersID)), *)});
      }
    }
    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, tsc_RB2, 
                                  cr_UEInformationResponse_LoggedMeasReportDef(v_AbsoluteTimeInfo, 
                                                                               v_PLMN_Identity, 
                                                                               v_LogMeasInfoList)));  //@sic R5s160060 sic@
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.1, Step 12");
  }


After:
	function fl_TC_8_6_2_1_Body() runs on UTRAN_PTC
  {
    var float v_WaitToAllowToActivateLogging := 5.0;
    var B28_Type v_CellId;
    var B48_Type v_AbsoluteTimeInfo;
    var UTRAN_SecurityInfo_Type     v_SecurityInfo;
    var UTRAN_CellInfo_Type v_CellInfo;
    var PLMN_Identity v_PLMN_Identity;
    var UTRAN_FDD_TDD_Type v_UTRAN_FDD_TDD_Type := f_IRAT_GetUTRAN_FDD_TDD_Mode();
    var CellParametersID v_CellParametersID;
    var UTRAN_CellInfo_Type_FDD v_CellInfoFDD;
    var template (present) LoggedMeasInfoList v_LogMeasInfoList;
    v_CellId := f_UTRAN34_GetCellIndentity(utran34_Cell1);
    v_SecurityInfo := f_UTRAN_Security_Get();
    v_CellInfo := f_UTRAN_CellInfo_Get(utran34_Cell1);
    v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_Cell1);
    v_AbsoluteTimeInfo := f_MDT_GetAbsolute_TimeStamp();
    //@siclog "Step 1" siclog@
    //SS transmits a LoggedMeasurementConfiguration message to configure the UE to perform logging of measurement results while in Cell_PCH.
    U_AM.send(cas_RLC_Data_Req_NoCnf(utran_CellDedicated, tsc_RB2, cs_LoggingMeasurementConfigurationDef(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                                                                         v_PLMN_Identity,
                                                                                                         v_AbsoluteTimeInfo)));
    //@siclog "Step 2,3" siclog@
    //Procedure to release RRC Connection
    f_UTRAN34_RRC_ConnRel(utran34_Cell1, cell_Dch);
    //@siclog "Step 4" siclog@
    //Wait 5 seconds for UE logging interval timer to expire at least once
    f_Delay(v_WaitToAllowToActivateLogging);
    //@siclog "Step 5" siclog@
    //SS changes cell power levels accroding to T2 in the Table 8.6.2.1-1
    f_UTRAN_SetCellPower(utran34_Cell1, -69, -69);
    f_UTRAN_SetCellPower(utran34_Cell2, -60, -60);
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_Cell2,utran34_Cell1/*WA#8_6_2_1*/); //@sic R5s160060 sic@
    //@siclog "Step 6" siclog@
    //The UE re-selects to Cell2 and performs Routing area update procedure. After completion of this procedure the UE is brought to idle mode
    f_UTRAN34_GMM_RAU(utran34_Cell2,?); // @sic R5s160060, R5-160925 sic@
    // Release RRC connection with cell2
    //f_UTRAN34_RRC_ConnRel(utran34_Cell2, cell_Dch); @sic R5s160060 sic@
    //@siclog "Step 7-10B" siclog@
    //Page UE to trigger RRC Connectin establishment
    //Complete RRC Connection setup and verify if IE "Logged Meas Available" is present
    f_UTRAN34_PS_MTCallEst(utran34_Cell2, true); //@sic R5s160060 sic@
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 10");    //@sic R5s160060 sic@
    //@siclog "Step 11" siclog@
    //Send UE INFORMATION REQUEST message
    v_SecurityInfo := f_UTRAN_Security_Get();
    U_AM.send(cas_RLC_Data_Req_NoCnf(utran_CellDedicated, tsc_RB2, cs_UEInformationRequest_LoggedMeasReportRequest(v_SecurityInfo.dl_IntegrityCheckInfo)));
    //@siclog "Step 12" siclog@
    // Receive UE INFORMATION RESPONSE message with Logged Meas Report
    select (v_UTRAN_FDD_TDD_Type) {
      case (UTRAN_FDD) {
        v_CellInfoFDD := f_UTRAN_CellInfo_GetFDDSpecific(utran34_Cell2);
        v_LogMeasInfoList := cr_LogMeasInfoList_FDD({permutation (cr_LogMeasInfo_FDD(complement(0),
                                                                                     cr_LoggedMeasServingCellMeas_FDD(v_CellId),
                                                                                     cr_LoggedMeasNeighbourInfoList_FDD(cr_PrimaryCPICH_Info(v_CellInfoFDD.priScrmCode))), *)});
      }
      case (UTRAN_TDD) {
        v_CellParametersID := f_UTRAN_CellInfo_GetCellParametersID_TDD(utran34_Cell2);
        v_LogMeasInfoList := cr_LogMeasInfoList_TDD128({permutation (cr_LogMeasInfo_TDD128(complement(0),
                                                                                           cr_LoggedMeasServingCellMeas_TDD128(v_CellId),
                                                                                           cr_LoggedMeasNeighbourMeasList_TDD128(v_CellParametersID)), *)});
      }
    }
    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, /*tsc_RB2 WA#8_6_2_1*/ tsc_RB4, 
                                  cr_UEInformationResponse_LoggedMeasReportDef(v_AbsoluteTimeInfo, 
                                                                               v_PLMN_Identity, 
                                                                               v_LogMeasInfoList, 
                                                                               * /*WA#8_6_2_1*/)));  //@sic R5s160060 sic@
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.2.1, Step 12");
  }


2.3 f_UTRAN34_GoToState6_7_MO() 
	Object name 
	f_UTRAN34_GoToState6_7_MO()

	Reason for change
	In the current TTCN implementation of this function, the expected QoS parameters that are passed to function , the function f_UTRAN_PS_SessionAct_ReceivePDP_Req are calculated earlier in function f_TC_8_6_2_1_UTRAN-> f_UTRAN34_PreambleGoToState6_10_MO-> f_UTRAN34_UE_PsRB_Est where the assignment v_QualityOfService := f_UTRAN_InitialiseQoS(p_CellId, v_RB_Configuration); is done.

However, the function f_UTRAN_InitialiseQoS is not setting the QoS parameters correctly. It uses the template cs_QoS_InteractiveOrBackgroundMT to initialise the QoS for configuration state cell_DCH_64kPS_RAB_SRB. The contents of this template are not as per the spec (e.g. setting of transferdelay IE), and moreover this template is for a MT case, not a MO situation like this. This causes a problem in the testcase when the UE transmits the PDP context and the testcase fails because the QoS parameters don’t match.

	Summary of change
	In the function f_UTRAN34_GoToState6_7_MO, created  a variable v_ExpectedQualityOfService and assigned it based on values of pc_AT_SupportToInit_PS_Call and pc_GERAN.

Passed this variable as a parameter in the function f_UTRAN_PS_SessionAct_ReceivePDP_Req.

Please see screenshot.  

	Module
	UTRAN34_PDP_Functions.ttcn

	Label
	WA#


Before:
	function f_UTRAN34_GoToState6_7_MO(UTRAN_CellId_Type p_CellId,
                                     template (value) QualityOfService p_QualityOfService,
                                     integer p_RBId := tsc_RB25,
                                     charstring p_APN := "",
                                     UTRAN_PDPContextTrigger_Type p_TriggerUE := TRIGGER_UE)
    runs on UTRAN_IP_PTC return ACTIVATEPDPCONTEXTREQUESTul
  {
    var template (value) EstablishmentCause v_EstCause := f_GetMoEstCause();
    var RB_ConfigType v_CellConfiguration := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    var ACTIVATEPDPCONTEXTREQUESTul v_ActPdpContextReq;
    var PDN_Index_Type v_PdnIndex := PDN_1;
    var boolean v_NewKeyFlag := true;
    var integer v_DedicatedId := 1;    
    if (p_TriggerUE == TRIGGER_UE) { // @sic R5s120097, R5s160060, R5-160730 sic@
      //Initiate an outgoing PS call
      f_UT_ActivatePDPContextUG(UT, v_DedicatedId, p_QualityOfService, p_APN); // @sic R5s160060 sic@
    }
    if (v_CellConfiguration == cell_DCH_64kPS_RAB_SRB)  {
      v_PdnIndex := PDN_2;
    }
    //Steps 1-3 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
    f_UTRAN34_RRC_ConnEst(p_CellId, v_EstCause);
    // Step 4 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
    f_UTRAN_GMM_ServReq(); // @sic R5s110465, R5s150838 sic@
    //Perform NAS session activation procedure for packet switched sessions (34.108 cl. 7.4.2.4.2)
    //Steps 1-2 in 34.108 cl. 7.4.2.4.2: Generate authentication and run authentication procedure
    f_UTRAN_GMM_Authentication();
    // Activate DHCP and ICMPv6 @sic R5-125676 sic@
    f_IP_Handling_Start(IP, v_PdnIndex, cs_DrbInfo_UTRAN(utran_CellDedicated, p_RBId)); // @sic R5s160060 sic@
    //Step 3-4 in 34.108 cl. 7.4.2.4.2: Transmit Security Mode Comamnd & Receive Security Mode Comlpete
    f_UTRAN34_RRC_Security(p_CellId, v_NewKeyFlag, ps_domain);
    //Step 5 in 34.108 cl. 7.4.2.4.2: Receive PDP Context Activation Request
    v_ActPdpContextReq := f_UTRAN_PS_SessionAct_ReceivePDP_Req(p_QualityOfService);
    return v_ActPdpContextReq;// We need to get TI from here (v_ActPdpContextReq.ti)
  }


After:
	function f_UTRAN34_GoToState6_7_MO(UTRAN_CellId_Type p_CellId,
                                     template (value) QualityOfService p_QualityOfService,
                                     integer p_RBId := tsc_RB25,
                                     charstring p_APN := "",
                                     UTRAN_PDPContextTrigger_Type p_TriggerUE := TRIGGER_UE)
    runs on UTRAN_IP_PTC return ACTIVATEPDPCONTEXTREQUESTul
  {
    var template (value) EstablishmentCause v_EstCause := f_GetMoEstCause();
    var RB_ConfigType v_CellConfiguration := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    var ACTIVATEPDPCONTEXTREQUESTul v_ActPdpContextReq;
    var PDN_Index_Type v_PdnIndex := PDN_1;
    var boolean v_NewKeyFlag := true;
    var integer v_DedicatedId := 1;
    var template (present) QualityOfService v_ExpectedQualityOfService;//WA#8_6_2_1
    if (p_TriggerUE == TRIGGER_UE) { // @sic R5s120097, R5s160060, R5-160730 sic@
      //Initiate an outgoing PS call
      f_UT_ActivatePDPContextUG(UT, v_DedicatedId, p_QualityOfService, p_APN); // @sic R5s160060 sic@
    }
    if (v_CellConfiguration == cell_DCH_64kPS_RAB_SRB)  {
      v_PdnIndex := PDN_2;
    }
    //Steps 1-3 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
    f_UTRAN34_RRC_ConnEst(p_CellId, v_EstCause);
    // Step 4 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
    f_UTRAN_GMM_ServReq(); // @sic R5s110465, R5s150838 sic@
    //Perform NAS session activation procedure for packet switched sessions (34.108 cl. 7.4.2.4.2)
    //Steps 1-2 in 34.108 cl. 7.4.2.4.2: Generate authentication and run authentication procedure
    f_UTRAN_GMM_Authentication();
    // Activate DHCP and ICMPv6 @sic R5-125676 sic@
    f_IP_Handling_Start(IP, v_PdnIndex, cs_DrbInfo_UTRAN(utran_CellDedicated, p_RBId)); // @sic R5s160060 sic@
    //Step 3-4 in 34.108 cl. 7.4.2.4.2: Transmit Security Mode Comamnd & Receive Security Mode Comlpete
    f_UTRAN34_RRC_Security(p_CellId, v_NewKeyFlag, ps_domain);
    //WA#8_6_2_1: Change start
    if (pc_AT_SupportToInit_PS_Call) {
      if (pc_GERAN) {
        v_ExpectedQualityOfService := cr_QoS_InteractiveOrBackgroundMO(p_QualityOfService.delayClass, p_QualityOfService.trafficClass);
      } else {
        v_ExpectedQualityOfService := cdr_QoS_InteractiveOrBackgroundMO_UMTS(p_QualityOfService.delayClass, p_QualityOfService.trafficClass);
      }
    } else {
      v_ExpectedQualityOfService := cr_QualityOfService_Any;
    }
    //WA#8_6_2_1 : Change stop
    //Step 5 in 34.108 cl. 7.4.2.4.2: Receive PDP Context Activation Request
    v_ActPdpContextReq := f_UTRAN_PS_SessionAct_ReceivePDP_Req(v_ExpectedQualityOfService/*WA#8_6_2_1 p_QualityOfService*/);
    return v_ActPdpContextReq;// We need to get TI from here (v_ActPdpContextReq.ti)
  }


